

# Lec4 VLSI 电路仿真

天津大学微电子学院

程思璐

[silu.cheng@tju.edu.cn](mailto:silu.cheng@tju.edu.cn)

# 主要内容


- 网表生成
- 工作点
- 传输函数分析
- 压控电压源
- 理想运算放大器
- 子电路
- 直流分析
- 绘制I-V曲线
- 双环直流分析
- 瞬态分析
- SIN源
- RC应用实例
- 交流分析
- 十倍频程和倍频程
- 分贝

- 脉冲语句
- 有限脉冲上升时间
- 阶跃响应
- RC电路的延迟与上升时间
- 分段线性源
- 仿真开关
- 电容的初始化条件
- 电感的初始化条件
- LC回路的Q值
- 理想积分器的频率响应
- 单位频率响应
- 积分器的时域性
- 收敛性
- 一些常见的错误和有用的技巧

## 4.1 生成网表

- 可由windows的记事本或写字板程序生成
- SPICE文件扩展名为: **.cir, .sp或.spi**
- 保存路径时不能出现中文路径


## 4.1 生成网表



**Figure 1.9** Saving a text file with a ".cir" extension.

## 4.2 工作点

- `.op`命令进行工作点分析
- 工作点仿真的输出数据不是图形，而只是节点电压、环路电流和小信号交流参数列表



```
*** Figure 1.11 CMOS ***
*#destroy all
*#run
*#print all
.op
Vin Vin 0 DC 1
R1 Vin Vout 1k
R2 Vout 0 2k
.end
```

**Figure 1.11** Operation point simulation for a resistive divider.

## 4.2 工作点

```
*** Figure 1.11 CMOS ***
*#destroy all
*#run
*#print all
.op
Vin Vin 0 DC 1
R1 Vin Vout 1k
R2 Vout 0 2k
.end
```

- 输入网表文件的第一个语句必须是标题行，不执行
- 最后一个语句必须是.END语句
- 以\*开头为注释行，可加在文件中的任何地方
- 以\*#开头的是命令行，多用于控制

## 4.2 工作点

```
*** Figure 1.11 CMOS ***
```

```
*#destroy all
```

//破坏先前的仿真数据

```
*#run
```

//运行仿真

```
*#print all
```

//打印仿真输出数据

```
.op
```

//执行工作点分析

```
Vin Vin 0 DC 1
```

//在节点Vin和地之间加直流电压源Vin=1V，电压源的名字必须以V开头

```
R1 Vin Vout 1k
```

```
R2 Vout 0 2k
```


//在Vin和Vout节点间加入1k电阻，在Vout节点和地之间加2k电阻。电阻的名字必须以R开头，地通常为0节点

```
.end
```

## 4.2 工作点

- 输出结果如下
- $V_{in} = 1.000000e + 00$
- $V_{out} = 6.666667e - 01$
- $V_{in} \# \text{ branch} = -3.333333e - 04$

//流经 $V_{in}$ 的电流。spice定义电流的正方向是从电压源的正端流向负端，因此电流值为负



## 4.3 传输函数分析 .TF


- **.TF**用于分析直流输入和输出电阻以及直流传输特性

例如 `.TF V(Vout,0) Vin`

- 每一次电路模拟过程中只能用一个**.TF** 语句，若输入文件中有多**.TF** 语句，则**HSPICE** 只执行最后一个**.TF** 语句

## 4.3 传输函数分析

- 在电路中串联一个0V电压源，是用于测量流经元件电流的常用方法。



```
*** Figure 1.12 CMOS ***
*#destroy all
*#run
*#print all
.TF I(Vmeas) Vin
Vin Vin 0 DC 1
R1 Vin Vout 1k
R2 Vout Vmeas 2k
Vmeas Vmeas 0 DC 0
.end
```

**Figure 1.12** Measuring the transfer function in a resistive divider when the output variable is the current through  $R2$  and the input is  $V_{in}$ .

## 4.3 传输函数分析

- 该.TF分析中，输出变量定义为电流 $I(V_{meas})$ ，输入定义为电压 $V_{in}$
- 仿真结果如下：
- $\text{Transfer\_function} = 3.333333e - 04$   
 $//I(V_{meas})/ V_{in}$
- $V_{in} \# \text{input\_impedance} = 3.000000e + 03$   
 $//\text{输入电阻}$
- $V_{meas} \# \text{output\_impedance} = 1.000000e + 20$   
 $//\text{输出电阻，此时输出为开路（}V_{meas}\text{从电路中去除了）}$

## 4.4 压控源


VCVS的  
控制节点

VCVS的  
输出节点

压控电压源(VCVS)定义以E为开头  
如下电路进行传输

增益


E1 Vt Vb Vin 0 23



**Figure 1.13** Example using a voltage-controlled voltage source.

## 4.5 理想运算放大器

- SPICE中可以用VCVS或压控电流源VCCS来实现一个理想运算放大器
- 流经节点n3和n4的电流等于节点n1和n2间的电压乘以跨导G。



Voltage-Controlled Current Source (VCCS)

G1 n3 n4 n1 n2 G

**Figure 1.14** Voltage-controlled current source in SPICE.

## 4.5 理想运算放大器

- 一个实例电路实现的理想运算放大器。

开环增益 $1 \times 10^6$ ，闭环增益-3




Figure 1.15 An op-amp simulation example.

## 4.6 子电路

- 当需要多次用到同一个电路单元时，可以生成一个子电路，在主体中调用。
- 子电路以.ends结尾，并要在.end前
- 调用子电路以X开头
- 子电路定义中的任何节点是局部变量，不能为接地点（节点为0）
- 子电路定义中的节点号、器件名、MODEL 的说明可以和外部的相同，而不会冲突。

## 4.6 子电路

- 定义子电路

```
.subckt Ideal_op_amp Vout Vp Vm
```

```
// Ideal_op_amp子电路参考名
```

```
Vout Vp Vm子电路外部节点
```

```
G1 Vout 0 Vm Vp 1MEG
```

```
//G为压控电流源的开头
```

```
RL Vout 0 1
```

```
.ends
```





Figure 1.15 An op-amp simulation example.

## 4.6 子电路

- 调用子电路
- X1 Vout 0 vm Ideal\_op\_amp



**Figure 1.15** An op-amp simulation example.

## 4.7 直流分析

- 直流(DC)分析中，输入为变量，对电路节点电压和电流进行仿真。绘制节点电压或电流图形。
- .dc Vin 0 1 1m  
表示Vin从0变化到1V，步长为1mV





Figure 1.16 DC analysis simulation for a resistive divider.

## 4.8 绘制I-V特性曲线

- .dc常用于有源器件的IV特性分析



```
*** Figure 1.17 CMOS ***
*#destroy all
*#run
*#let ID=-Vin#branch
*#plot ID
.dc Vin 0 1 1m
Vin Vin 0 DC 1
R1 Vin Vd 1k
D1 Vd 0 mydiode
.model mydiode D
.end
```




Figure 1.17 Plotting the current-voltage curve for a diode.

## 4.8 绘制I-V特性曲线

- 定义二极管以D开头
- .model参数D代表二极管模型
- SPICE中定义通过电压源的电流方向为正





Figure 1.17 Plotting the current-voltage curve for a diode.

```
*** Figure 1.17 CMOS ***
*#destroy all
*#run
*#let ID=-Vin#branch
*#plot ID
.dc Vin 0 1 1m
Vin Vin 0 DC 1
R1 Vin Vd 1k
D1 Vd 0 mydiode
.model mydiode D
.end
```

## 4.8 绘制I-V特性曲线

仿真结果：



**Figure 1.17** Plotting the current-voltage curve for a diode.

## 4.9 双环直流分析

- 在直流分析中加入一个外部回路，同时控制两个变量




Figure 1.18 Plotting the current-voltage curves for an NPN BJT.

.DC VDS 0 10 0.5 VGS 0 5 1

## 4.10 瞬态分析

- 瞬态分析在时域对电路进行仿真（x轴为时间）
- 瞬态分析语句形式为：

`.tran tstep tstop, <tstart><tmax><uic>`


- tstep——时间步长
- tstop——仿真结束时间
- tstart——仿真开始时间
- tmax——最大时间步长

## 4.10 瞬态分析

- 网表:


\*\*\*Figure 1.19 CMOS:Circuit Design,  
Layout,and Simulation\*\*\*

```
*#destroy all  
*#run  
*#plot vin vout  
.tran 100p 100n  
Vin Vin 0 DC 1  
R1 Vin Vout 1k  
R2 Vout 0 2k  
.end
```



## 4.10 瞬态分析

- 仿真结果:



**Figure 1.19** Transient simulation for the circuit in Fig. 1.11.

## 4.11 SIN源


- 产生正弦波语句:

**SIN Vo Va freq<td> <theta>**

- Vo——正弦曲线偏置电压（与正弦波叠加的直流电压）
- Va——正弦波振幅
- freq——正弦波的频率
- td——仿真中正弦波出现前的延迟
- theta——正弦波的幅度衰减特性

## 4.11 SIN源

SIN语句用于瞬态仿真分析，不可应用于交流分析



\*\*\* Figure 1.20 \*\*\*

```
*#destroy all
```

```
*#run
```

```
*#plot vin vout
```


```
.tran 1n 3u
```

```
Vin Vin 0 DC 0 SIN 0 1 1MEG
```

```
R1 Vin Vout 1k
```


```
R2 Vout 0 2k
```

```
.end
```



**Figure 1.20** Simulating a resistive divider with a sinusoidal input.

## 4.12 RC电路实例



\*\*\* Figure 1.21 \*\*\*

\*#destroy all

\*#run

\*#plot vin vout


.tran 10u 30m

Vin Vin 0 DC 0 SIN 0 1 200

R1 Vin Vout 1k


CL Vout 0 1u

.end



**Figure 1.21** Simulating the operation of an RC circuit using a .tran analysis.

## 4.13 另一种RC电路实例



\*\*\* Figure 1.22 \*\*\*

\*#destroy all

\*#run

\*#plot vin vout

.tran 10u 30m

Vin Vin 0 DC 0 SIN 0 1 200

R1 Vin Vout 1k

C1 Vin Vout 2u

C2 Vout 0 1u

.end




Figure 1.22 Another RC circuit example.

## 4.14 交流分析

- .ac分析确定电路的频率响应（x轴为频率）

`.ac dec nd fstart fstop`

- dec——x轴为十倍频程形式，可替换为lin（线性），oct（倍频程形式）
- nd——每十倍频程的点数（如100）
- fstart,fstop——起始和截止频率

注意：fstart不能是0或者DC，因为不是交流信号；

## 4.14 交流分析

\*\*\*Figure 1.23 CMOS: Circuit Design, Layout, and Simulation

```
*#destroy all
```

```
*#run
```

```
*#plot db(vout/vin)
```

```
*#set units=degrees
```

```
*#plot ph(vout/vin)
```


```
.ac dec 100 1 10k //交流分析
```

```
Vin vin 0 DC 0 SIN 0 1 200 AC 1 //交流分析中SIN定义的语句
```

```
R1 vin vout 1k
```

```
CL vout 0 1u
```


```
.end
```



被忽略掉,不执行,且必须增加交流源。其中**AC 1**即为幅值或峰值为1的交流小信号源。

## 4.14 交流分析

仿真结果:



**Figure 1.23** AC simulation for the RC circuit in Fig. 1.21.

## 4.15 十倍频程和倍频程

- 十倍频程：乘以或除以10

23MHz之上的十倍频程是230MHz

1.2kHz之下的十倍频程是120Hz

- 倍频程：乘以或除以2

23MHz之上的倍频程是46MHz

1.2kHz之下的倍频程是600Hz

23MHz之上的两个倍频程是92MHz




Figure 1.23 AC simulation for the RC circuit in Fig. 1.21.

## 4.16 分贝

- 幅值响应减小为原来的 $1/10$ ，等价于减小了 $-20\text{dB}$   
( $20 \times \log(1/10)$ )
- 幅值响应增加为原来的 $10$ 倍，等价于上升了 $20\text{dB}$   
( $20 \times \log(10)$ )
- 幅值响应以 $-40\text{dB}/\text{十倍频程}$ 下降表示频率每增大 $10$ 倍，幅值下降为 $1/100$
- 幅值响应以 $-6\text{dB}/\text{倍频程}$ 下降表示频率每增大 $2$ 倍，幅值下降为幅值响应的 $1/2$

## 4.17 脉冲语句

- 瞬态仿真中定义脉冲或时钟信号  
`pulse vinit vfinal td tr tf pw per`
- vinit,vfinal——初始和终值电压
- td——脉冲开始前的延迟时间
- tr,tf——脉冲的上升和下降时间（这两个参数为0将使用瞬态仿真中的步长）
- pw——脉冲宽度
- per——脉冲周期

## 4.17 脉冲语句

- `Vin Vin 0 DC 0 pulse 0 1 6n 0 0 3n 10n`
- 上例中输入脉冲的上升和下降时间不是0，而是瞬态语句中建议最大步长100ps





Figure 1.24 Simulating the step response of an RC circuit using a pulsed source voltage.

## 4.18 有限脉冲上升时间

- 改: `Vin Vin 0 DC 0 pulse 0 1 6n 10p 10p 3n 10n`



**Figure 1.25** Specifying a rise time in the pulse statement to avoid slow rise times (rise times set by the maximum step size in the .tran statement.)

## 4.19 阶跃响应

- 利用脉冲语句也可以产生一个阶跃函数

`Vin Vin 0 DC 0 pulse 0 1 2n 10p`

- 减小延迟时间，制定上升时间。不设定脉冲宽度，脉冲跃迁后在仿真范围内保持高电平，形成RC电路的阶跃响应。





Figure 1.26 Step response of an RC circuit.




## 4.20 RC电路延迟与上升时间

- RC电路中脉冲达到终值50%时所用时间:

$$T_d \approx 0.7RC$$

- 上升或下降时间为:


$$T_r \approx 2.2RC$$



**Figure 1.26** Step response of an RC circuit.

## 4.20 RC电路延迟与上升时间

- 在上一电路中输入-1V跃迁到-2V的脉冲
- `Vin Vin 0 DC 0 pulse -1 -2 2n 10p`




**Figure 1.27** Another step response (negative going) of an RC circuit.

## 4.21 分段线性源

- 分段线性源（PWL）可以指定任意波形  
`Pwl t1 v1 t2 v2 t3 v3...<rep>`
- t1时刻电压为v1，t2时刻电压为v2.....
- 例：Pwl 0 0.5 3n 1 5n 1 5.5n 0 7n 0

## 4.21 分段线性源



在0到3ns之间，输入电压线性变化

**Figure 1.28** Using a PWL source to drive an RC circuit.



## 4.22 仿真开关

- 一个电压控制的电阻代表一个基本的开关特征
- 节点2和节点0之间的电阻线性地从10meg变化到1m欧姆，当节点1和0之间的电压从0变化到1的时候。超过了电压限制，阻抗仍然分别在10meg和1m欧姆。

Gswitch 2 0 VCR PWL(1) 1 0 0v, 10meg 1v, 1m

Gxxx n+ n- VCCAP PWL(1) in+ in- <DELTA=val> <SCALE=val> <M=val>  
+ <TC1=val><TC2=val> x1, y1 x2, y2 ... x100, y100 <IC=val>  
+ <SMOOTH=val>

## 4.23 电容的初始化条件



\*\*\* Figure 1.30 \*\*\*

```
*#destroy all
*#run
*#plot vout
```

```
.tran 100p 8n UIC
```

\\**.tran**后有**uic**表示使用“初始化条件”或跳过初始工作点计算

```
Vclk clk 0 pulse -1 1 2n
```

```
Vin Vin 0 DC 5
```

```
S1 Vin Vouts clk 0 switmodel
```

```
R1 Vouts Vout 1k
```

```
C1 Vout 0 1p IC=2
```

\\**IC=2** 为电容初始化电压，置于电容定义之后


```
.model switmodel sw ron=0.1
```

```
.end
```

\\设定一个节点电压语句  
**.ic v(vout) = 2**

**Figure 1.30** Using initial conditions and a switch in an RC circuit simulation.

## 4.24 电感的初始化条件



\*\*\* Figure 1.31 \*\*\*

```
*#destroy all
*#run
*#plot vout

.tran 100p 8n UIC

Vclk clk 0 pulse -1 1 2n
Vin Vin 0 DC 5
S1 Vin Vouts 0 clk switmodel
R1 Vouts Vout 1k
R2 Vout 0 1k
L1 Vout 0 10u IC=5m


.model switmodel sw ron=0.1

.end
```


**Figure 1.31** Using initial conditions in an inductive circuit.

## 4.25 LC回路的Q值

- 谐振电路品质因数 $Q$
- 品质因数定义为回路中存储的能量与损耗的能量的比值
- $Q$ : 中心（谐振）频率与3dB响应的带宽之比
- 例：中心频率503MHz，上3dB为511.2MHz，下3dB为494.8MHz  
 $Q$ 值约为30



## 4.25 LC回路中的Q值



\*\*\* Figure 1.32 \*\*\*

```
*#destroy all
*#run
*#plot db(vout)

.AC lin 100 400MEG 600MEG

lin Vout 0 DC 0 AC 1
R1 Vout 0 1k
L1 Vout 0 10n
C1 Vout 0 10p

.end
```

**Figure 1.32** Determining the Q, or quality factor, of an LC tank.

## 4.26理想积分器中的频率响应

- 流过电阻的电流必然与流过电容的电流相等，所以

$$\frac{V_{in}}{R} + \frac{V_{out}}{1/j\omega C} = 0$$

- 或者

$$\frac{V_{in}}{V_{out}} = \frac{-1}{j\omega RC} = \frac{-(1 + j \cdot 0)}{0 + j\omega RC}$$


- 积分器传输函数的幅值为

$$\left| \frac{V_{in}}{V_{out}} \right| = \frac{\sqrt{(-1)^2 + (-0)^2}}{\sqrt{(0)^2 + (2\pi f RC)^2}} = \frac{1}{2\pi RC f}$$

- 而通过积分器的相移

$$\angle \frac{V_{out}}{V_{in}} = \tan^{-1} \frac{-0}{-1} - \tan^{-1} \frac{2\pi RC f}{0} = -90^\circ$$


- 注意积分器的增益在频率减小至直流时接近无穷大，而相移为常数



## 4.27 单位频率增益



- 单位增益频率  $f_{un}$  —— 传输函数幅值为1时的频率。

$$\left| \frac{V_{out}}{V_{in}} \right| = 1 = \frac{1}{2\pi RC f_{un}} \rightarrow f_{un} = \frac{1}{2\pi RC}$$



## 4.27 单位频率增益

带入原理图中的值，单位增益频率为**159Hz**，spice仿真如下



\*\*\* Figure 1.33 \*\*\*

```
*#destroy all
*#run
*#plot db(vout/vin)
*#set units=degrees
*#plot ph(vout/vin)

.ac dec 100 1 10k

Vin Vin 0 DC 1 AC 1
Rin Vin vm 1k
Cf Vout vm 1u

X1 Vout 0 vm Ideal_op_amp
.subckt Ideal_op_amp Vout Vp Vm
G1 Vout 0 Vm Vp 1MEG
RL Vout 0 1
.ends
.end
```

Figure 1.33 An integrator example.

## 4.28积分器的时域特性

- 积分器时域特性通过电阻电流等于电容电流可知：

$$V_{out} = \frac{1}{C} \int \frac{V_{in}}{R} dt$$

- 如果电压恒定，输出线性上升或下降，即：

$$V_{out} = \frac{V_{in}}{RC} \int dt$$

- 该式可用来设计将方波转换为锯齿波的电路。

## 4.28积分器的时域特性




Figure 1.34 Time-domain integrator example.

\*\*\* Figure 1.34 \*\*\*

```
*#destroy all
*#run
*#plot vout vin
```

```
.tran 10u 10m
.ic v(vout)=0
```

```
Vin Vin 0 DC 1
+ pulse -1 1 0 1u 1u 2m 4m
Rin Vin vm 1k
Cf Vout vm 1u
```

```
X1 Vout 0 vm Ideal_op_amp
.subckt Ideal_op_amp Vout Vp Vm
G1 Vout 0 Vm Vp 1MEG
RL Vout 0 1
.ends
.end
```

+表示该行为是上一行代码的继续，有顺序性

## 4.29 收敛性

- 不收敛的网表可以调节三个参数帮助收敛：

ABSTOL、VNTOL、RELTOL

- ABSTOL——绝对电流容限，默认为1pA  
当结果在真实值附近1pA范围之内时，SPICE认为电流收敛
- VNTOL——节点电压容限，默认为1uV
- RELTOL——相对容限参数，默认为0.1%
- RELTOL与前两个同时使用时，较大的值生效
- 为增强收敛可以添加如下语句：  
.OPTIONS ABSTOL=1uA VNTOL=1mV RELTOL=0.01
- **注：**增加这三个参数将加速仿真和收敛，代价是降低精度。

## 4.30 常见的错误和有用的技巧

(1) **SPICE**网表中第一行必须是标题。**SPICE**会忽略网表文件中的第一行。

(2) 使用**1MEG**定义1兆欧，而不是**1M**或**1m**。

(3) 用**1**表示1法拉，而不是**1f**或者**1F**。**1F**意味着千万分之一或者 $10^{-15}$ 法拉。

(4) 电压源命名应以字母**V**开头。电流源命名应以字母**I**开头。

(5) 在**.tran**语句后面增加**UIC**（使用初始化条件）常常有助于解决瞬态仿真的收敛问题。

## 4.30 常见的错误和有用的技巧

(6) 瞬态仿真显示时间数据，即X轴是时间轴。一个不光滑平整的图形，比如看起来像三角正弦波，或者仅是不光滑的图形，是由于没有确定最大打印步长的结果。

(7) 除非采用实际（最终）尺寸确定宽度和长度，否则在使用MOSFET的仿真中必须在.options语句中包含比例因子。

(8) 通常，PMOS器件的衬底区连接VDD，NMOS的体区连接地。

(9) 避免电源边界通常有助于直流扫描的收敛。例如，从0到1V的扫描，电路可能不收敛，但从0.05到0.95扫描将会使电路收敛。

(10) 在任何仿真中增加.OPTION RSHUNT = 1 E8（或者某些其他电阻值）可有助于收敛。该语句在电路中的每个节点都增加一个并联电阻。使用太小的值会影响仿真结果。